Variations on the Sequence of Apéry Numbers

Frédéric Chyzak

(Version of January 9, 1998)

In the early 1990's, Doron Zeilberger and Herbert Wilf, developped a new methodology for symbolic summation and integration (Wilf, Herbert S. and Zeilberger, Doron (1992):An algorithmic proof theory for hypergeometric (ordinary and ''q'') multisum/integral identities, *Inventiones Mathematicae*, **108**:575-633). One of the most famous successes of this so-called ''WZ-method'' has been to provide a computer proof of the combinatorial identity

$$\sum_{k=0}^{n} \mathrm{binomial}(n, k)^2\, \mathrm{binomial}(n+k, k)^2 =$$

$$\sum_{k=0}^{n} \mathrm{binomial}(n, k)\, \mathrm{binomial}(n+k, k) \left(\sum_{j=0}^{k} \mathrm{binomial}(k, j)^3 \right)$$

and to prove that the sequence of these numbers a_n satisfies the second order recurrence equation

$$(n+2)^3\, u_{n+2} - ((n+2)^3 + (n+1)^3 + 4\,(2\,n+3)^3)\, u_{n+1} + (n+1)^3\, u_n = 0.$$

Proving this recurrence was a crucial step of Apéry's proof for the irrationality of

$$\zeta(3) = \sum_{k=1}^{\infty} \frac{1}{k^3}.$$

On the other hand, the identity itself stems from a number-theoretic question raised by Schmidt in (Schmidt, Asmus L. (1990): Generalized Legendre polynomials, *J. reine angew. Math.*, **404** :192-202).

Several proofs of the identity above, that relates the Apéry numbers a_n to the Franel numbers

$$f_n = \sum_{k=0}^{n} \mathrm{binomial}(n, k)^3,$$

were given in (Strehl, Volker (1994): Binomial Identities, Combinatorial and Algorithmic Aspects, *Discrete Math.*, **136**:309-346). One of them in particular is based on Zeilberger's algorithm for hypergeometric summation, and yields the recurrence equation above as a by-product. In the following sections, we first recall how Apéry was led to the identity, borrowing from Van der Poorten's report (Van der Poorten, Alfred (1979): A Proof that Euler missed... Apéry's Proof of the Irrationality of $\zeta(3)$, *Math. Intelligencer*, **1**:195-203); we next give a proof for both results using our package Mgfun, and finally exploit the recurrence equation to beat *Maple* computing many digits of $\zeta(3)$.

Sketch of Apéry's Proof

Apéry's first remark is that the double sequence

$$c_{n,k} = \left(\sum_{m=1}^{n} \frac{1}{m^3}\right) + \left(\sum_{m=1}^{k} \frac{(-1)^{(m+1)}}{2\, m^3 \operatorname{binomial}(n, m) \operatorname{binomial}(n+m, m)}\right)$$

tends to $\zeta(3)$ uniformly in k for k in $(1 .. \infty)$ when n tends to infinity. This stems from the alternating series being uniformly bounded by $\frac{1}{n}$. However, the convergence of this series is not strong enough so as to show the irrationality of $\zeta(3)$. Apéry used summation methods to accelerate the convergence. Namely, define

$$a_n = \sum_{k=0}^{n} \operatorname{binomial}(n, k)^2 \operatorname{binomial}(n+k, k)^2\, c_{n,k} \quad \text{and}$$

$$b_n = \sum_{k=0}^{n} \operatorname{binomial}(n, k)^2 \operatorname{binomial}(n+k, k)^2,$$

then $\frac{a_n}{b_n}$ also tends to $\zeta(3)$. Here appears the crucial recurrence of Apéry: one remarks that it is satisfied by both sequences a and b, with initial conditions

$$a_0 = 0, \quad a_1 = 6, \quad \text{and} \quad b_0 = 1, \quad b_1 = 5.$$

By a number-theoretic argument, it follows from this recurrence that

$$\zeta(3) - \frac{a_n}{b_n} = \mathrm{O}({q_n}^{(-1+\delta)}),$$

with

$$q_n = 2 \operatorname{lcm}(1, 2, \ .. , n)^3\, b_n \quad \text{and} \quad \delta = \frac{4 \ln(1+\sqrt{2}) - 3}{4 \ln(1+\sqrt{2}) + 3}, \text{ which is positive.}$$

This is sufficient to prove that $\zeta(3)$ is irrational, and yields an irrationality measure of at least

$$1 + \frac{1}{\delta} = 13.417820 \,..$$

Proof of Apéry's Recurrence

In this session, we chiefly use the user-oriented package Mgfun.

```
> with(Mgfun);
```

$$[\mathit{diag\_of\_sys}, \mathit{int\_of\_sys}, \mathit{pol\_to\_sys}, \mathit{sum\_of\_sys}, \mathit{sys}{*}\mathit{sys}, \mathit{sys}{+}\mathit{sys}]$$

Recurrence for the Left-Hand Side

We first prove that the Apéry numbers, as defined by the left-hand side

$$a_n = \sum_{k=0}^{n} \operatorname{binomial}(n, k)^2 \operatorname{binomial}(n+k, k)^2,$$

satisfy the announced recurrence.

The summand

> $f := \operatorname{binomial}(n, k)^2 \operatorname{binomial}(n+k, k)^2$

satisfies both following equations:

```
> h(n+1,k)/h(n,k)=factor(normal(subs(n=n+1,f)/f,expanded));
```

$$\frac{h(n+1,k)}{h(n,k)} = \frac{(n+1+k)^2}{(-n-1+k)^2}$$

```
> h(n,k+1)/h(n,k)=factor(normal(subs(k=k+1,f)/f,expanded));
```

$$\frac{h(n,k+1)}{h(n,k)} = \frac{(n+1+k)^2(-n+k)^2}{(k+1)^4}$$

This yields the following system

```
> sys:=collect(map(numer,map(eq->op(1,eq)-op(2,eq),{"",""})),
  h);
```

$$sys := \{\\ (n^2+2n-2nk+1-2k+k^2)h(n+1,k)+(-2nk-1-n^2-2n-2k-k^2)h(n,k)\\ ,(-n^4-2n^3-n^2+2n^2k+2n^2k^2+2nk+2nk^2-k^2-2k^3-k^4)h(n,k)\\ +(k^4+4k^3+6k^2+4k+1)h(n,k+1)\}$$

where each element $expr$ in the set denotes the equation $expr = 0$. The definite summation over k in $(0, n)$ is performed by the following call to Mgfun[sum_of_sys]:

```
> sum_of_sys(sys,k=-infinity..infinity,takayama_algo);
```

$$[(n^3+3n^2+3n+1)h(n)+(-34n^3-231n-153n^2-117)h(n+1)\\ +(n^3+6n^2+12n+8)h(n+2)]$$

```
> rec[left]:=op(collect(",h,factor));
```

$$rec_{left} := (n+1)^3 h(n) + (n+2)^3 h(n+2) - (2n+3)(17n^2+51n+39)h(n+1)$$

This is the announced recurrence in disguise.

Recurrence for the Inner Sum

We next compute the right-hand side, beginning with the Franel numbers f_n given by

$$f_n = \sum_{k=0}^{n} \mathrm{binomial}(n,k)^3.$$

We first obtain a system to describe the summand:

$> f := \mathrm{binomial}(n,k)^3$

$> \dfrac{h(n+1,k)}{h(n,k)} - \mathrm{factor}\left(\mathrm{normal}\left(\dfrac{\mathrm{subs}(n=n+1,f)}{f}, expanded\right)\right)$

$> \dfrac{h(n,k+1)}{h(n,k)} - \mathrm{factor}\left(\mathrm{normal}\left(\dfrac{\mathrm{subs}(k=k+1,f)}{f}, expanded\right)\right)$

```
> sys:=collect(map(numer,{"",""}),h);
```

$$sys := \{(-n^3+3n^2k-3nk^2+k^3)h(n,k)+(k^3+3k^2+3k+1)h(n,k+1),\\ (-n^3-3n^2+3n^2k-3n+6nk-3nk^2-1+3k-3k^2+k^3)h(n+1,k)\\ +(n^3+3n^2+3n+1)h(n,k)\}$$

Summing, we get a recurrence for the Franel numbers f_n.

```
> sys:=sum_of_sys(sys,k=-infinity..infinity,natural_boundari
  es);
```

$$sys := \{(-56-136n-104n^2-24n^3)h(n)+(3n^3+22n^2+51n+36)h(n+3)$$

$$+(-45n^3-240-240n^2-419n)\,\mathrm{h}(n+1)$$
$$+(-18n^3-148-114n^2-232n)\,\mathrm{h}(n+2)\}$$

System for the Right-Hand Product

Let us multiply the f_k by binomial(n, k) binomial($n+k, k$). To do so, we prepare two systems of recurrence equations, one for each factor.

The following system describes f_k, which is independent from n.

```
> sys1:={h(n+1,k)-h(n,k)} union eval(subs(h=proc(k) h(n,k)
  end,subs(n=k,sys)));
```

$$sys1 := \{\mathrm{h}(n+1,k)-\mathrm{h}(n,k),\ (-56-136k-104k^2-24k^3)\,\mathrm{h}(n,k)$$
$$+(3k^3+22k^2+51k+36)\,\mathrm{h}(n,k+3)+(-45k^3-240-240k^2-419k)\,\mathrm{h}(n,k+1)$$
$$+(-18k^3-148-114k^2-232k)\,\mathrm{h}(n,k+2)\}$$

We next obtain a system that describes binomial(n, k) binomial($n+k, k$), the weight to multiply with:

> $f :=$ binomial(n, k) binomial($n+k, k$)

> $$\frac{\mathrm{h}(n+1,k)}{\mathrm{h}(n,k)}-\mathrm{factor}\left(\mathrm{normal}\left(\frac{\mathrm{subs}(n=n+1,f)}{f},\ expanded\right)\right)$$

> $$\frac{\mathrm{h}(n,k+1)}{\mathrm{h}(n,k)}-\mathrm{factor}\left(\mathrm{normal}\left(\frac{\mathrm{subs}(k=k+1,f)}{f},\ expanded\right)\right)$$

```
> sys2:=collect(map(numer,{"",""}),h);
```

$$sys2 := \{(-n^2-n+k+k^2)\,\mathrm{h}(n,k)+(k^2+2k+1)\,\mathrm{h}(n,k+1),$$
$$(-n-1+k)\,\mathrm{h}(n+1,k)+(n+1+k)\,\mathrm{h}(n,k)\}$$

We finally perform the product by a call to Mgfun['sys*sys']:

```
> sys:=`sys*sys`(sys1,sys2);
```

$$sys := \{(-n-1-k)\,\mathrm{h}(n,k)+(n+1-k)\,\mathrm{h}(n+1,k),\ (-672k+672n-272k^6$$
$$+224n^2+2304nk-2528k^2+1608n^2k+2664n^2k^2+3048nk^2-3816k^3$$
$$-2960k^4+600k^4n^2+72k^5n-280n^4-840n^3+1944k^3n+72n^5k+56n^6$$
$$-384n^4k^2-1368n^3k+168n^5+24n^6k-768n^3k^2-624n^4k-72k^3n^4$$
$$-144k^3n^3+600k^4n-24k^7+1872k^3n^2-1248k^5+72k^5n^2)\,\mathrm{h}(n,k)+(2880$$
$$+11748k-1920n+600k^6-1680n^2-5272nk+20132k^2-4853n^2k$$
$$-5512n^2k^2-5752nk^2+18817k^3+10372k^4-840k^4n^2-90k^5n+240n^4$$
$$+480n^3-3118k^3n+240n^4k^2+838n^3k+480n^3k^2+419n^4k+45k^3n^4$$
$$+90k^3n^3-840k^4n+45k^7-3073k^3n^2+3374k^5-90k^5n^2)\,\mathrm{h}(n,k+1)+($$
$$-17384k^2-1798k^5+1532n^2k^2+18k^5n^2+18k^5n+760k^3n^2+1532nk^2$$
$$+1520n^2k+186k^4n+760k^3n+592n+1520nk-18k^7-276k^6-13740k^3$$
$$+592n^2-12080k-3552-6448k^4+186k^4n^2)\,\mathrm{h}(n,k+2)$$
$$+(-5184k^2-52k^6-3k^7-3675k^3-1540k^4-3996k-1296-382k^5)\,\mathrm{h}(n,k+3)$$
$$\}$$

Recurrence for the Right-Hand Side

We obtain a recurrence for the right-hand side by applying Mgfun[sum_of_sys] on the system computed in the previous section.

```
> sum_of_sys(sys,k=-infinity..infinity,takayama_algo);
```

$$[(-112-466n-775n^2-295n^4-655n^3-6n^6-67n^5)\,h(n) + (37042n^3+14280+204n^6+13611n^4+2612n^5+55685n^2+43966n)\,h(n+1) + (6n^6+113n^5+870n^4+3495n^3+7700n^2+8784n+4032)\,h(n+4) + (-88200-92178n^3-189205n^2-204n^6-3508n^5-202734n-24811n^4)\,h(n+3) + (-57976n^3-77840-140445n^2-167086n-11775n^4-942n^5)\,h(n+2)]$$

This recurrence is different from the one obtained for the left-hand side.

```
> rec[right]:=op(collect(",h,factor));
```

$$rec_{right} := -(2n+7)(2n+3)(n+3)(51n^3+469n^2+1418n+1400)\,h(n+3) + (2n+3)(3n+7)(n+3)(n+4)^3\,h(n+4) - (3n+8)(2n+7)(n+2)(n+1)^3\,h(n) - (2n+5)(471n^4+4710n^3+17213n^2+27190n+15568)\,h(n+2) + (2n+7)(2n+3)(n+2)(51n^3+296n^2+553n+340)\,h(n+1)$$

We thus need more work to prove that both sides agree.

Final Proof of the Identity and of the Second Order Recurrence

Let h_n be any solution of the second order recurrence which has been obtained for the left-hand side:

```
> rec[left];
```

$$(n+1)^3\,h(n) + (n+2)^3\,h(n+2) - (2n+3)(17n^2+51n+39)\,h(n+1)$$

Thus:

```
> h(n+2)=collect(solve(",h(n+2)),h,normal);
```

$$h(n+2) = -\frac{(n^3+3n^2+3n+1)\,h(n)}{n^3+6n^2+12n+8} + \frac{(34n^3+153n^2+231n+117)\,h(n+1)}{n^3+6n^2+12n+8}$$

```
> subs(n=n+1,");
```

$$h(n+3) = -\frac{((n+1)^3+3(n+1)^2+3n+4)\,h(n+1)}{(n+1)^3+6(n+1)^2+12n+20} + \frac{(34(n+1)^3+153(n+1)^2+231n+348)\,h(n+2)}{(n+1)^3+6(n+1)^2+12n+20}$$

and

```
> subs(n=n+1,");
```

$$h(n+4) = -\frac{((n+2)^3+3(n+2)^2+3n+7)\,h(n+2)}{(n+2)^3+6(n+2)^2+12n+32} + \frac{(34(n+2)^3+153(n+2)^2+231n+579)\,h(n+3)}{(n+2)^3+6(n+2)^2+12n+32}$$

Then, h_n also solves the recurrence for the right-hand side:

```
> collect(subs(",""","""",rec[right]),h,normal);
```

$$0$$

At this point, we have proved that both sides of the equation satisfy the same recurrence of order 4. To prove the announced equality, we simply need to check 4 initial conditions, since the leading coefficient of the recurrence of order 4,

```
> coeff(rec[right],h(n+3));
```

$$-(2\,n+7)\,(2\,n+3)\,(n+3)\,(51\,n^3+469\,n^2+1418\,n+1400)$$

never vanishes for non-negative n. Now the proof of the identity

$$> \; eq := \sum_{k=0}^{n} \mathrm{binomial}(n,k)^2\,\mathrm{binomial}(n+k,k)^2 =$$
$$\sum_{k=0}^{n} \mathrm{binomial}(n,k)\,\mathrm{binomial}(n+k,k)\left(\sum_{j=0}^{k} \mathrm{binomial}(k,j)^3\right)$$

simplify follows from

```
> eval(subs(n=0,Sum=add,eq));
```

$$1 = 1$$

```
> eval(subs(n=1,Sum=add,eq));
```

$$5 = 5$$

```
> eval(subs(n=2,Sum=add,eq));
```

$$73 = 73$$

```
> eval(subs(n=3,Sum=add,eq));
```

$$1445 = 1445$$

Therefore, the Apéry numbers also satisfy the announced second order recurrence.

Computation of ζ(3) Using Standard Maple and the Holonomic Approach

Standard Maple

Maple has numerical routines for almost all special functions it knows about. Here is the corresponding calculation for $\zeta(3)$.

```
> ti[0]:=time():
> Z3[standard]:=evalf(Zeta(3),332);
```

$Z3_{standard}$:= 1.2020569031595942853997381615114499907649862923404988817922\
2715553418382057863130901864558736093352581461991577952607194184919\
5998673283213776396837207900161453941782949360066719191575522242494\
4396156390966410329115909578096551465127991840510571525598801543710\
7811020398275325667876035223369849416618110570147157786394997375237\
53

```
> ti[standard]:=time()-ti[0];
```

$$ti_{standard} := 38.315$$

Holonomic Approach

We compute an approximation of $\zeta(3)$ using Apéry's recurrence. More precisely, we compute it as $\frac{a_{200}}{b_{200}}$. (Remember that $\frac{a_n}{b_n}$ tends to $\zeta(3)$.)

```
> ti[0]:=time():
> N:=200:
```

To do so, we use the gfun package by Salvy and Zimmermann (Salvy, Bruno and Zimmermann, Paul (1994): Gfun: a Maple package for the manipulation of generating and holonomic functions in one variable, *ACM Trans. Math. Software*, **20**(2):163-177).

```
> with(gfun);
```

[*Laplace, algebraicsubs, algeqtodiffeq, algeqtoseries, algfuntoalgeq, borel, cauchyproduct, diffeq*diffeq, diffeq+diffeq, diffeqtohomdiffeq, diffeqtorec, guesseqn, guessgf, hadamardproduct, holexprtodiffeq, invborel, listtoalgeq, listtodiffeq, listtohypergeom, listtolist, listtoratpoly, listtorec, listtoseries, listtoseries/Laplace, listtoseries/egf, listtoseries/lgdegf, listtoseries/lgdogf, listtoseries/ogf, listtoseries/revegf, listtoseries/revogf, maxdegcoeff, maxdegeqn, maxordereqn, mindegcoeff, mindegeqn, minordereqn, optionsgf, poltodiffeq, poltorec, ratpolytocoeff, rec*rec, rec+rec, rectodiffeq, rectohomrec, rectoproc, seriestoalgeq, seriestodiffeq, seriestohypergeom, seriestolist, seriestoratpoly, seriestorec, seriestoseries*]

The *gfun* package provides us with a routine for transforming a recurrence equation like

$$> \; eq := n^3\,\mathrm{u}(n) - (34\,n^3 - 51\,n^2 + 27\,n - 5)\,\mathrm{u}(n-1) + (n-1)^3\,\mathrm{u}(n-2)$$

into a procedure. Each of the following procedures encodes the calculation of a sequence u_n given by the equation *eq* and its initial values u_0 and u_1.

```
> A:=rectoproc({eq,u(0)=0,u(1)=6},u(n));
```

A := proc(*n*)
local *i, u0, u1, u2*;
    u0 := 0;
    u1 := 6;
    for *i* **from** 2 **to** *n* − 1 **do**
        u2 := − (−*u0* + 5**u1*
            + (3**u0* − 27**u1* + (−3**u0* + 51**u1* + (*u0* − 34**u1*)**i*)**i*)**i*) / *i*^3;
        u0 := *u1*;
        u1 := *u2*
    od;
    − (−*u0* + 5**u1* + (3**u0* − 27**u1* + (−3**u0* + 51**u1* + (*u0* − 34**u1*)**n*)**n*)**n*) /
        n^3
end

(0) = 0

#(1)=6

```
> B:=rectoproc({eq,u(0)=1,u(1)=5},u(n));
```

B := **proc**(*n*)

local *i*, *u0*, *u1*, *u2*;

u0 := 1;

u1 := 5;

for *i* **from** 2 **to** *n* − 1 **do**

$u2 := -(-u0+5*u1$
$+(3*u0-27*u1+(-3*u0+51*u1+(u0-34*u1)*i)*i)*i)\ /\ i\hat{}3;$

u0 := *u1*;

u1 := *u2*

od;

$-(-u0+5*u1+(3*u0-27*u1+(-3*u0+51*u1+(u0-34*u1)*n)*n)*n)\ /$
$n\hat{}3$

end

#(0)=1

#(1)=5

Compute a_{200} and b_{200}:

```
> a:=A(N);
```

a := 27120214525404796485818105490387356973722905867369270689468377839\
44224561403894128396087948809726922826015017836044109623384971164259\
58401545583169451199761039685134669488602519840929656495220226753418\
96089773180958620255598124721366216659076995569703855978725017570960\
73139780572287415078593525290640230547195432404735627435195518652493\
99645197675211923855456207500790113208841509085283616289600327339645\
26988349617205476424290099324103833143545597684684919761345878795570\
61056249835054669010795802078038561593537842855077584521632237323882\
732096779955723555299 / 174889846834181722648388614514189413474473\
33892895988949119209188831918537107959961683147172089686841121791959\
49160149943568467894775048501378397003271807504496214678596660937743\
45063238004074842236955459054744393441648002194104941464018656046547\
25939200000000

```
> b:=B(N);
```

b := 12900409496639264260168205845938654727954502873043001846622201173\
42540619222381480791908924782479089649848693303030274561021434381123\
74509927628755291501618734616451402070459237474365982171435137805480\
47395759213403223437590088956423380519683504906353460506233080769659\

$4554733147636155936662958465207425$

A priori, it is not clear how many digits we can guarantee.

```
> P:=round(evalf(ln(a)/ln(10))*1.1);
```

$$P := 332$$

```
> Z3[holonomy]:=evalf(a/b,P);
```

$Z3_{holonomy} := 1.202056903159594285399738161511449990764986292340498881792\backslash$
$271555341838205786313090186455873609335258146199157795260719418491 99\backslash$
$599867328321377639683720790016145394178294936006671919157552224249 42\backslash$
$439615639096641032911590957809655146512799184051057152559880154371 09\backslash$
$781102039827532566787603522336984941661811057014715778639499737523 78\backslash$
53

The time used is

```
> ti[holonomy]:=time()-ti[0];
```

$$ti_{holonomy} := 7.286$$

Comparison

The holonomic approach is several time faster.

```
> ti[standard]/ti[holonomy];
```

$$5.258715344$$

Moreover, this ratio would increase with the accuracy of the calculations. In this session, we have obtained $\zeta(3)$ up to 332 digits.

```
> Z3[standard]-Z3[holonomy];
```

$$0$$

However, *Maple* would be able to compute $\zeta(z)$ for *any* complex value z, while the holonomic approach only works for $z = 3$.